
About FractMovie:


( Real-time generation of multiple 2-d/3-d IFS fractals


( User-definable time-variant IFS parameters for animation 


( User-definable camera movement to zoom/pan/rotate 


( User-definable ambient/point light sources 


( Export in Compuserve GIF or Windows BMP format





Who uses it:


( Those who like to create stylish fractal images for their web pages. 


( Those who like to play with 2-d/3-d IFS fractals but would not bother to write a program. 


( Those who like to explore fractal animations





On what platform:


( IBM compatible PC with intel 486DX or above, 4Mb RAM, and VESA compatible super VGA for 256 colour 640x480 resolution mode. 


( MS DOS 5.0 or above.

Chapter 1 





INSTALL FractMovie





This chapter describes how to install and run FractMovie.











1.1 SYSTEM REQUIREMENTS





FractMovie runs on IBM PC compatible with Intel 80486DX or above. The program requires 4Mb RAM and SVGA for VESA 640x480 256 colour mode. To obtain smoother movies, a Pentium is recommended.





The current version (V1.62) works with MS-DOS 5.0 or above. You may be able to run it from a DOS window. If you cannot do so, you can exit the windows and restart your PC in the real-mode DOS. Win 95 users can click the “Real-mode DOS” icon or click the “Start” icon at the bottom of the screen and choose “Reboot to real-mode DOS” from the menu.








1.2 INSTALLATION





To install FractMovie on your hard drive, first make a directory for your FractMovie and then copy to this directory all the files unzipped from the downloaded zip file fm.zip. After installation, your FractMovie directory should have the following files:


�



File name�
Description�
�
FM.BAT�
batch file to start FractMovie�
�
FM_.EXE�
executable�
�
DOS4GW.EXE�
DOS pretected-mode utility�
�
README.TXT�
brief information about FractMovie�
�
MANUAL.DOC�
user’s manual (MS word)�
�
MANUAL.PS�
user's manual (Postscript)�
�
ANIM1.FMS�
script: animated pyramid 1�
�
ANIM2.FMS�
script: square to fern�
�
ANIM3.FMS�
script: animated pyramid 2�
�
ANIM4.FMS�
script: evaporating fern�
�
ANIM5.FMS�
script: filling Sierpinsky gasket�
�
ANIM6.FMS�
script: magic triangle�
�
SCENE1.FMS�
script: pyramid surrounded by trees�
�
SCENE2.FMS�
script: tree in garden�
�
SCENE3.FMS�
script: ferns on Sierpinsky gasket�
�
SCENE4.FMS�
script: clouds on carpet�
�






1.3 MODIFY THE DOS PATH





If you like to run FractMovie from any directory, you will need to alter your DOS path to include the FractMovie directory. You probably set your DOS path with a PATH statement in your AUTOEXEC.BAT file--- if so, just append the FractMovie directory to that statement. Your PATH statement will end up looking something like the one that follows





	PATH C:\DOS;C:\WINDOWS;C:\FM





The path modification will not take effect until you restart your PC.








1.4 QUICK START





Your FractMovie package has included ten fractal movie scripts for you to begin with. You can type at the DOS prompt, for example,





	C:\>FM SCENE1.FMS





to see a fractal pyramid surrounded by four trees, or





	C:\>FM ANIM1





for an animated fractal pyramid. The file extension “.FMS” is omitted as this is the default extension.








1.5 COMMAND-LINE OPTIONS





FractMovie has a number of command-line options which allow the user to take snap shots, export to files, change the window size, and to display the time.








1.51 Snap Shot





Option -Stime is used to display a snapshot at the time specified by time. If you type





	C:\>FM ANIM1.FMS -S10





your PC screen will show a still picture at � EMBED Equation.2  ���. 








1.52 Export to File





You can use option -Ofile to export snap shots to files in Compuserve GIF or in Windows BMP format. The following command





	C:\>FM ANIM1.FMS -S10 -OTRY.GIF





for example, will store the snap shot at � EMBED Equation.2  ��� to a GIF file TRY.BMP








1.53 Set Movie Window Size





You can set the movie window size by using the -wSize option. The window size is measured in relative linear units: the full screen is 1, one with half width and height is 0.5 and so on. If you like to set the window to the full screen size, for example, you can use  following command





	C:\>FM ANIM1.FMS -w1





A larger window displays more details while a smaller window runs faster.





�
1.62 Display the Time





You can choose to display or not to display the time by using the -t[+|1] option. If you choose ‘+’ for yes, the time will be displayed beneath the rendering window. The format of the displayed time is two digits for hours followed by two digits for minutes which is followed by another two digits for seconds.








�
Chapter 2





CREATING FRACTAL MOVIES





This chapter describes how to make IFS fractal movies. The idea is to construct scenes using IFS fractals which can be painted with color patterns and be placed at various places. The scene is then illuminated and a camera with a defined movement trajectory is used to shoot the movie script. The whole process is done on the computer. All we need is to write a fractal movie script to specify the parameters and run FractMovie to show the movie on your PC screen.








2.1 QUICK START





We start with a motion picture that zooms into Barnsley's fern. The fractal fern is generated by an IFS of four contractive affine transforms:





	� EMBED Equation.2  ���





	� EMBED Equation.2  ���


	


	� EMBED Equation.2  ���





	� EMBED Equation.2  ���





The first step to make the zoom-into-the-fern movie is to define the above IFS:





�
	IFS fern2d {


	  0.04,  -0.004,  0,


	  0.004,  0.18,   0;


	  0.85,   0.04,   0,


	 -0.04,   0.85,   1.6;


	  0.2,   -0.26,   0, 


	  0.23,   0.22,   1.6;


	 -0.15,   0.28,   0,


	  0.26,   0.24,   0.44;


	}





The key word IFS marks the beginning of a construct for defining an IFS. It is followed by an identifier fern2d so that the defined IFS can later be used to generate fractal objects. The parameters of the four transforms are enclosed in a pair of braces. The order in which they appear is consistent with the mathematical equation.





The next step is to place a copy of the fractal generated by a defined IFS in the scene. We use the place construct:





	place fern2d {


	  colour  0, 1, 0;


	}





The key word place signals the starting of a construct that puts in the scene a copy of the fractal generated by a defined IFS. The following identifier fern2d is the name of the defined IFS. The colour statement enclosed in the braces set the color of the whole fern in RGB format (0, 1, 0) which is green.





Now that there is one fractal in the scene we illuminate the scene with uniform ambient lighting:





	ambient  0.5, 0.5, 0.5;





The triple (0.5, 0.5, 0.5) following the key word ambient is the intensity of the ambient light in RGB format.





We place the camera at (x,y,z)=(0, 0, 30) facing the x-y plane where the fern lies and 


reduce with time the distance between the camera and the x-y plane to achieve the zoom-in effect. This is done with the camera statement:





	camera  0, 0, 30*pow(2,-0.1*t),  0, 0, 0;





The first three parameters after the keyword camera set the camera position to be � EMBED Equation.2  ���. The z coordinate � EMBED Equation.2  ��� is time-varying: z=30 at t=0 and z gradually approaches 0 as t increases. The next three parameters are directional angles. The triple (0, 0, 0) makes the camera point to the negative z axis.





The zoom-into-the-fern movie is to last one minute, i.e. to start from t=0 and to stop at t=60. The start and stop statements are used for this purpose:





	start 0;


	stop  60;





Puting them all together, the completed script for the zoom-into-to-fern movie is as follows





	IFS fern2d {


	  0.04,  -0.004,  0,


	  0.004,  0.18,   0;


	  0.85,   0.04,   0,


	 -0.04,   0.85,   1.6;


	  0.2,   -0.26,   0, 


	  0.23,   0.22,   1.6;


	 -0.15,   0.28,   0,


	  0.26,   0.24,   0.44;


	}


	


	place fern2d {


	  colour  0, 1, 0;


	}


	


	ambient  0.5, 0.5, 0.5;





	camera  0, 0, 11*pow(2,-0.1*t),  0, 0, 0;


	start  0;


	stop  60;





Assuming that the above script has been saved to a file FERN.FMS (provided with the FractMovie package for your convenience), we can type 





	C:\FM>FM FERN.FMS 





to see the zoom-into-the-fern movie. We observe that finer and finer details of the fractal fern are revealed as the camera moves closer to it. How many leaves does the fern have? Infinite. We have created an object of infinite details from a few dozens of parameters!








2.2 TWO- AND THREE- DIMENSIONAL LINEAR IFS





A fractal is a mathematical object that is self-similar. That is, the whole is made up of a number of smaller copies of itself. The relation between the whole and a smaller version of itself can be described by a transform. Therefore a fractal is adequately described by the transforms that map the whole to the smaller copies of itself. In mathematical terms, this group of transforms are called an Iterated Function System (IFS).





A transform may have various properties. If the transforms that make the IFS are affine, that is, the transforms map lines to lines, the IFS is often termed as a linear IFS. Since the constituent parts of a fractal are smaller than the whole, these transforms are shrinking mappings, that is, such a transform maps any pair of points to a pair of points whose distance is smaller than that of the original pair.





Fractals in the Cartesien plane, or 2-d fractals, can be described by IFS of 2-d affine transforms. A 2-d affine transform takes the form:





	� EMBED Equation.2  ���


	


or, in matrix format,





	� EMBED Equation.2  ���





where � EMBED Equation.2  ���,� EMBED Equation.2  ���,� EMBED Equation.2  ���, and � EMBED Equation.2  ���.





A 2-d affine transform is determined by its 6 parameters or “codes”. An IFS, therefore, can also be represented as a table of codes. A table of codes for the Barnsley fern, for example, is given below





�
a11�
a12�
a13�
a14�
b1�
b2�
�
w1�
0.04�
-0.006�
0.006�
0.18�
0�
0�
�
w2�
0.85�
0.04�
-0.04�
0.85�
0�
1.6�
�
w3�
0.2�
-0.26�
0.23�
0.22�
0�
1.6�
�
w4�
-0.15�
0.28�
0.26�
0.24�
0�
0.44�
�



We often see in the literature a probability assigned to each transform. The probabilities are calculated according to that (1) the sum of the probabilities is 1; (2) the probability of a transform is proportional to the absolute value of  the determinant of the transform matrix. The probabilities are required by the Random Algorithm to generate the attractor. They are, however, not needed by some other IFS generation algorithms, e.g. the Deterministic Algorithm. The probabilities are not considered here as an inherent part of the IFS.





The Fractal Movie Script Language provides a easy way to define an IFS. The definition starts with the key word IFS and an identifier for the IFS. The identifier enables the defined IFS to be later used to generate fractal copies. The transform codes are enclosed in a pair of curly braces ‘{‘ and ‘}’. The order in which the 6 parameters of a transform appear is the same as they appear in the equation. The parameters are separated with a ‘,’ within a transform. The last parameter ends with a ‘,’. The syntax is





	IFS name {


	label:


	  a11,  a12,   b1,


	  a21,  a22,   b2;


	         .


	         .


	         .


	}





We often like to identify the constituent parts that make the fractal in order to assign different colors to different parts. This can be done by attaching a label to each of the transform. For example, we attach four labels to the four transforms that generate the four constituent parts of the fractal fern:





	IFS fern2d {


	stem:


	  0.04,  -0.004,  0,


	  0.004,  0.18,   0;


	upper:  


	  0.85,   0.04,   0,


	 -0.04,   0.85,   1.6;


	left: 


	  0.2,   -0.26,   0, 


	  0.23,   0.22,   1.6;


	right:  


	 -0.15,   0.28,   0,


	  0.26,   0.24,   0.44;


	}





The attractor generated by a 2-d IFS lies in a plane, or, in other words, is a subset of a plane. To generate fractals that have a “volume” (whose enclosing convex polyhedron has a volumn), we use 3-d IFS. A 3-d IFS consists of a number of 3-d affine transforms. A 3-d affine transform takes the form:





	� EMBED Equation.2  ���





or, in matrix format,





	� EMBED Equation.2  ���





The following IFS of five 3-d affine transforms generates a fractal that resembles a pyramid:





	� EMBED Equation.2  ���





	� EMBED Equation.2  ���


	


	� EMBED Equation.2  ���





	� EMBED Equation.2  ���


	


	� EMBED Equation.2  ���





The definition of a 3-d IFS in the Fractal Movie Script Language is similar to that of a 2-d IFS. The only difference lies in the number of parameters for each transform: 6 for a 2-d transform and 12 for a 3-d transform. The syntax for defining 3-d IFS is





	IFS name {


	label:


	  a11,  a12,  a13,   b1,


	  a21,  a22,  a23,   b2,


	  a31,  a32,  a33,   b3;


	         .


	         .


	         .


	}





Using the syntax shown above, the IFS for fractal pyramid can be defined as follows where the transform for the upper smaller pyramid is labeled as “upper” and those for 


the four smaller pyramids at the bottom are labeled as “a”, “b”, “c”, and “d”:





�
	IFS pyramid {


	top:


	  0.5,  0,    0,    0,


	  0,    0.5,  0,    1,


	  0,    0,    0.5,  0;


	a:  


	  0.5,  0,    0,    0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5,  0.5;


	b:  


	  0.5,  0,    0,   -0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5,  0.5;


	c:  


	  0.5,  0,    0,   -0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5, -0.5;


	d:  


	  0.5,  0,    0,    0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5, -0.5;


	}








2.3 CREATE FRACTAL SCENES





A fractal scene is made up with a number of IFS fractals. We have in the previous section described how to define an IFS and we now describe how to place the fractals generated by the defined IFS in the scene. To put in the scene a copy of the fractal generated by a defined IFS, we use the place construct:





	place IFS_name {


	  coordinate_transforms;


	  colorings;


	}





The place construct starts with the place and the name of an already-defined IFS. Two types of operations may be applied on the fractal: (1) the fractal can be moved, scaled, and rotated by using coordinate transforms; (2) different parts of the fractal can be painted with desired colors. The coordinate transform statements and coloring statements are enclosed in a pair of braces ‘{’ and ‘}’.








2.31 Coordinate Transforms





Coordinate transforms enable us to move, scale, and rotate a fractal generated by an IFS.


The coordinate transform operations include translation (sliding), scaling (stretching), and rotation with respect to any of the three coordinate axises.





To slide the fractal, we use





	translate <axis> amount;





where axis enclosed in the brackets is one of the three letters x, y, and z indicating the coordinate axis, amount is an expression specifies the amount of required translation. The following statement, for example, moves the fractal along the x axis by 10 units:





	translate <x> 10;





Scaling of a fractal is described by the scale statement





	scale <axis> factor;





where axis indicates the coordinate axis in which the scaling is to be performed. The expression factor specifies the scaling factor. To scale the fractal down to half of its original height, we perform a y-axis scaling with the scaling factor being 0.5:





	scale <y> 0.5;





Rotation is defined with the rotate statement





	rotate <axis> angle;





where axis specifies the coordinate axis around which the rotation is to be performed, the following expression angle set the rotation. The unit for measuring the angle is radiant.








2.32 Color Assignments





A fractal consists of a number of smaller copies of itself. Each such copy is identified with a string of transform label. We can assign any one of such copy a color. The color is determined by a triple (r, g, b) which is the ratios of the reflected light to the incoming light expressed in the RGB format. The triple (1, 0, 0) is color red since the red component of the incoming light is refracted to the viewer while the green and blue components are lost; the triple (0.5, 0.5, 0.5) is grey since half of the red, green and blue component of the incoming light is reflected to the viewer. A special color “.transparent” will make the region transparent, that is, as if the region does not exists.





To assign a color (red, green, blue) to a region identified with the transform label string TLS, we use the color statement 





	color <TLS> red, green, blue;





The following statement paints the whole fractal with color (red,green,blue).





	color <> red, green, blue;





The empty string in the above statement represents the entire fractal.





The color of a region is determined by the first color statement that assign a color to the region. The subsequent statements which also attempt to color the same region


will have no effect. The following sequence of color statements will paint the whole fractal in green as the second statement and the following have no effect.





	color <> 0, 1, 0;


	color <> 0, 0, 0;


	color <> .transparent;








2.4 ANIMATE THE FRACTALS





An IFS fractal is generated by an IFS. The shape of the fractal depends on the parameters of the transforms that make the IFS. We can set some of the parameters to change with time and hence obtain an animated fractal- a fractal that evolves with time.





One way to create an animated fractal is to start with a static fractal and let a number of parameters of the IFS be a function of time. For example, we can start with the following IFS which generates a fractal pyramid





�
	IFS pyramid {


	top:


	  0.5,  0,    0,    0,


	  0,    0.5,  0,    1,


	  0,    0,    0.5,  0;


	a:  


	  0.5,  0,    0,    0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5,  0.5;


	b:  


	  0.5,  0,    0,   -0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5,  0.5;


	c:  


	  0.5,  0,    0,   -0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5, -0.5;


	d:  


	  0.5,  0,    0,    0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5, -0.5;


	}





We now alter the parameters of the transform that generates the upper smaller copy to incorporate a rotation with time. The resultant IFS is





	IFS pyramid {


	top:


	  0.5*cos(0.075*t),     0.5*sin(0.075*t),  0,    0,


	  0.5*(-sin(0.075*t)),  0.5*cos(0.075*t),  0,    1,


	  0,                    0,                 0.5,  0;


	a:  


	  0.5,  0,    0,    0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5,  0.5;


	b:  


	  0.5,  0,    0,   -0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5,  0.5;


	c:  


	  0.5,  0,    0,   -0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5, -0.5;


	d:  


	  0.5,  0,    0,    0.5,


	  0,    0.5,  0,    0,


	  0,    0,    0.5, -0.5;


	}





This time-varing IFS generates an animated fractal pyramid. Type 





	C:\FM>FM ANIM1.FMS





to see the animation sequence. We observe that the introduced rotation operation not only applies to the top smaller pyramid but also to ... This results in a sophisticated shape evolution.





Another way to generate an animated sequence is to use “linear interpolation”. The idea is to first find the initial and final fractals, and then let the parameters linearly change with time from those of the initial IFS to those of the final IFS. That is





	� EMBED Equation.2  ���





where � EMBED Equation.2  ��� is a parameter of the initial IFS, � EMBED Equation.2  ��� is a corresponding parameter of the final IFS, k is a constant controling the speed of animation. The duration of the animation is 1/k. For k=0.1, for example, we have p=p1 at t=0 and p gradually approaches p2 and we have p=p2 at t=10.





We now create an animation sequence to show a square evoling into a fractal fern.The starting point is an IFS for the square:





	� EMBED Equation.2  ���





	� EMBED Equation.2  ���


	


	� EMBED Equation.2  ���	


	


	� EMBED Equation.2  ���





The final point is an IFS for the fractal fern:





	� EMBED Equation.2  ���





	� EMBED Equation.2  ���


	


	� EMBED Equation.2  ���





	� EMBED Equation.2  ���





We set k=0.04 and let the animation last 25 seconds. The parameters of the resultant square-to-fern animation are calculated with Equation~\ref{equ:anim}. The following is a FMSL script for the square-to-fern animation





	// square


	float s1a11=0.5;   float s1a12=0;      float s1b1=0;


	float s1a21=0;     float s1a22=0.5;    float s1b2=1;





	float s2a11=0.5;   float s2a12=0;      float s2b1=0;


	float s2a21=0;     float s2a22=0.5;    float s2b2=3;  





	float s3a11=0.5;   float s3a12=0;      float s3b1=-1;


	float s3a21=0;     float s3a22=0.5;    float s3b2=2;





	float s4a11=-0.5;  float s4a12=0;      float s4b1=1;


	float s4a21=0;     float s4a22=0.5;    float s4b2=2;





	// fern


	float f1a11=0.04;  float f1a12=-0.004; float f1b1=0;


	float f1a21=0.004; float f1a22=0.18;   float f1b2=0;





	float f2a11=0.85;  float f2a12=0.04;   float f2b1=0;


	float f2a21=-0.04; float f2a22=0.85;   float f2b2=1.6;





	float f3a11=0.2;   float f3a12=-0.26;  float f3b1=0;


	float f3a21=0.23;  float f3a22=0.22;   float f3b2=1.6;





	float f4a11=-0.15; float f4a12=0.28;   float f4b1=0;


	float f4a21=0.26;  float f4a22=0.24;   float f4b2=0.44;





	// time-varying paramters


	float t0=t/15;


	float t1=1-t0;


	


	float t1a11=t0*f1a11+t1*s1a11;


	float t1a12=t0*f1a12+t1*s1a12;


	float t1a21=t0*f1a21+t1*s1a21;


	float t1a22=t0*f1a22+t1*s1a22;


	float t1b1 =t0*f1b1 +t1*s1b1;


	float t1b2 =t0*f1b2 +t1*s1b2;





	float t2a11=t0*f2a11+t1*s2a11;


	float t2a12=t0*f2a12+t1*s2a12;


	float t2a21=t0*f2a21+t1*s2a21;


	float t2a22=t0*f2a22+t1*s2a22;


	float t2b1 =t0*f2b1 +t1*s2b1;


	float t2b2 =t0*f2b2 +t1*s2b2;





	float t3a11=t0*f3a11+t1*s3a11;


	float t3a12=t0*f3a12+t1*s3a12;


	float t3a21=t0*f3a21+t1*s3a21;


	float t3a22=t0*f3a22+t1*s3a22;


	float t3b1 =t0*f3b1 +t1*s3b1;


	float t3b2 =t0*f3b2 +t1*s3b2;





	float t4a11=t0*f4a11+t1*s4a11;


	float t4a12=t0*f4a12+t1*s4a12;


	float t4a21=t0*f4a21+t1*s4a21;


	float t4a22=t0*f4a22+t1*s4a22;


	float t4b1 =t0*f4b1 +t1*s4b1;


	float t4b2 =t0*f4b2 +t1*s4b2;





	// definition of the square-to-fern IFS


	IFS square2fern {


	  t1a11,  t1a12,  t1b1,


	  t1a21,  t1a22,  t1b2;


	  t2a11,  t2a12,  t2b1,


	  t2a21,  t2a22,  t2b2;


	  t3a11,  t3a12,  t3b1,


	  t3a21,  t3a22,  t3b2;


	  t4a11,  t4a12,  t4b1,


	  t4a21,  t4a22,  t4b2;


	}





	


2.5 LIGHTING THE SCENE





The display screen of a computer consists of an 2-d array of pixels. The color and brightness of each pixel can be represented with a triple (r,g,b) where r is the red component, g the green component, b the blue component. The intensity of each color component is measured on an increasing scale from 0 to 1. That is, 0 implies the zero intensity and 1 implies the maximum intensity. The triple (0,0,0) is black since all the three component intensity are zero.





If a pixel is on the background, that is, the pixel is not on any of the fractals, the brightness and color of the pixel is that of the background which is defined by the background statement:





	background red, green, blue;





The following statement set the background to be black:


 


	background 0, 0, 0;





If a pixel is on one of the fractals, the brightness and color of the pixel is the brightness and color of the corresponding point on the fractal, which depends on the lighting intensity at that point and the color of the point (the ratios at which the red, green, and blue components of the incoming light are reflected to the viewer).





Ambient lighting gives an even illumination to every fractal in the scene. The intensity is expressed in RGB format, and is specified with the following statement





	ambient r_amb,g_amb,b_amb;





For a point with color (r_color,g_color,b_color) under the above ambient lighting, the perceived light intensity (the color and brightness of the pixel) is (r_amb.r_color,g_amb.g_color,b_amb.b_color). The perceived color is jointly determined by the lighting and the (material) color of the point. If the color of the point is white and the lighting is red, then the perceived color of the point is red.








2.6 DESIGN THE CAMERA MOVEMENT





The zoom/pan/rotate effects are obtained via camera movement. To zoom into a fractal, we gradually reduce the distance between the camera and the fractal; To pan around a scene, we rotate the camera horizontally; To rotate the scene, we rotate the camera around its lens axis.





The location and attitude of a camera is determined by its three





The zoom-into-the-fern movie in Section~\ref{sec:qs} uses the following camera parameters














�
Chapter 3





THE RACTAL MOVIE SCRIPT


LANGUAGE





This chapter describes the Fractal Movie Script Language (FMSL)- a script language for making IFS fractal movies.








3.1 KEYWORDS





Keywords play a special grammatical role, that is, its occurrence marks the begining of a construct or statement. The following are the keywords:





	ambient 


	background 


	camera 


	color/colour 


	float


	IFS/ifs


	light 


	place 


	rotate 


	scale 


	start 


	stop 


	translate 


	visibility





Keywords are tokens that exactly match one of the above listed words. Those that contain keywords as substrings are not recognized as keywords. For example, none of the following is a keyword although they contain keywords as substrings





	colourful 


	place1 





FMSL is a case-sensitive language: upper case and lower case letters are treated as different. The following are not keywords although they share the same spelling with the keywords:





	Light


	LIGHT 


3.2 CONSTANTS





A constant can be an integer or a floating-point number. An integer is a sequence of digits optionally preceded by a minus sign. The following, for example, are integers:





	1 


	-1 


	12345





A floating-point number is a string, optionally preceded by a minus sign, consisting of a sequence of digits, a decimal point, another sequence of digits and optionally followed by the letter ‘e’ or ‘E’ and an integer exponent. The following, for example, are floating-point numbers:





	3.1415926 


	1.23456e12








3.3 VARIABLES





Variables in FMSL are used to store the value of an expression. They are functionally identical to the defining expression. A variable is defined with the statement:





	float name=expr;





where name is a name given to the variable, expr is an expression of constants and already-defined variables.





A variable name is a sequence of letters and digits starting with a letter. The following three tokens





	a 


	A 


	myFractal01





for example, are legal variable names. Notice that “a” and “A” are taken as two different variables since FMSL is a case-sensitive language.





After definition, a variable is associated with the defining expression. It cannot be redefined nor can its value be modified after the definition. The following is incorrect since the second statement attemps to modify the defined variable “a”:





	float a=sin(1);


	a=a+1; 





The predefined variable “t” holds the current time in seconds. That is, the value of the variable “t” is the current time measured in seconds.








3.4 ARITHMETIC OPERATORS





The arithmetic operators are ‘+’, ‘-’, ‘*’, and ‘/’. The binary ‘+’ and ‘-’ operators have the same precedence, which is lower than the precedence of the ‘*’ and  ‘/’ operators, which is in turn lower than the precedence of the unary ‘-’ operator. The arithmetic operators associate left to right.





The left to right association implies that the expression





	a/b/c





is interpreted as 





	(a/b)/c





Precedence plays an important role. The expression





	a+b*c





is interpreted as 





	a+(b*c)





since the ‘*’ operator has higher precedence. 








3.5 MATHEMATICAL FUNCTIONS





The following table lists the mathematical functions that are supported in FMSL:


�



�
Notation�
Name�
�
sqrt(x)�
square root of x�
�
pow(x,y)�
x to the power of y�
�
exp (x)�
exponent of x�
�
log(x)�
natural logrithm of x�
�
sin(x)�
sine of x�
�
cos(x)�
cosine of x�
�
tan(x)�
tangent of x�
�
asin(x)�
arch sine of x�
�
acos(x)�
arch cosine of x�
�
atan(x)�
arch tangent of x�
�
atan2(x,y)�
arch tangent of x/y�
�






3.6 DEFINING IFS





The definition of an IFS starts with the keyword IFS and a name for the IFS. An IFS name is a sequence of letters and digits starting with a letter. Following the name is the IFS body enclosed in a pair of braces, which includes the definitions of the transforms:





	IFS IFS_name {


	  transform1;


	  transform2;


	     .


	     .


	     .


	}





A transform definition has an optional label and a list of the parameters of the transform. For a 2-d affine transform as given below





	� EMBED Equation.2  ���





the definition in FMSL is:





	label:


	  expr11,expr12,expr13,


	  expr21,expr22,expr23;





For a 3-d transform:





	� EMBED Equation.2  ���





the definition, similarly, is:





	label:


	  expr11,expr12,expr13,expr14,


	  expr21,expr22,expr23,expr24,


	  expr31,expr32,expr33,expr34;








3.7 PLACING IFS FRACTALS IN THE SCENE





We have described how to define an IFS in the last section. This section describes how to put the defined IFS fractals into the scene with the place construct. The place construct starts with the keyword place and the name of an already-defined IFS to be used to generate a fractal object. Following the name is the construct body which may include translate, scale, rotate, and colour statements.





	place IFSname {


	  rotation;


	  scaling;


	  translation;


	  coloring;


	}





IFS fractals can be translated (shifted) along the axises. Translation of the IFS fractal is specified with the translate statement:





	translate  <axis> expr;





where axis enclosed in a pair of brackets is a letter- x, y, or z- specifying the coordinate axis along which the translation is performed, the expression expr set the amount of translation. The following example





	translate <y> 10.0;





will perform translation of 10 units along the y axis.





IFS fractals can also be scaled and rotated. Scaling and rotation are defined with the scale and rotate statements respectively:





	scale <axis> expr;


	rotate <axis> expr;





where axis indicates the coordinate axis for the operation, expr in the scale statement specifies the scaling factor, and expr in the rotate statement set the rotation angle in radiant.





Different parts of an IFS fractal can be assinged a different color using the color statemetn. A colour statement consists of the keyword colour, an address string enclosed in a pair of brackets, and the intended colour in RGB format:





	colour <address> expr1, expr2, expr3;





The address string is a sequence of transform labels, separated with white spaces. The question mark ‘?’ may be used as a match-all label. For example, if the three transforms generating the upper, left, and right constituent parts of the Sierpinsky gasket are labeled ‘u’, ‘l’, and ‘r’ respectively, the address string <u l> is the left portion of the upper constituent triangle while <? l> is all the three left portions of the three constituent triangles.








3.8 ILLUMINATING THE SCENE





A fractal scene can be illuminated with ambient lighting as well as point light sources. The ambient lighting component is defined with the ambient statement:





	ambient expr1,expr2,expr3;





where the three parameters expr1, expr2, and expr3 are respectively the red, green, and blue lighting intensity.





A point light source is created with the light statement:





	light expr1,expr2,expr3,expr4,expr5,expr6;





where the first three parameters expr1, expr2, and expr3 are the x, y, and z coordinates of the light source. The next three parameters expr4, expr5, and expr6 are the red, green, and blue intensity.








3.9 BACKGROUND COLOR





The background color of a scene is set by the statement 





	background expr1,expr2,expr3;





where the first, second, and third parameters are the red, green, and blue components of the background colour. The triple (0,0,0) represents black and (1,1,1) white.








3.10 MIST EFFECTS





Mist effects may provide further cues for the perception of distance. The farther the objects are away from the camera, the more vague they appear. They eventually vanish into the background as the distance tends to infinity. Mist effects is achieved with the statement 





	visibility expr;





where expr specifies the distance at which objects are “half” blurred.








3.11 CAMERA POSITION





The camera position is specified by its three coordinates and three rotation angles. The three coordinates are x, y, and z respectively and the three angles specify horizontal rotation, vertical tilt, and rotation around the lens axis. If the three angles are all zero, the camera points to -z and faces the x-y plane in the conventional position: x axis points to the right and y axis points upwards..





The syntax for specifying the camera position is 





	camera expr1,expr2,expr3,expr4,expr5,expr6;





The first three parameters expr1, expr2, and expr3 specify the x, y, and z coordinates, the next three parameters expr4, expr5, and expr6 specify the three rotation angles. 








3.12 STARTING AND STOPPING TIMES





The starting and stopping times of a fractal movie are defined with the start and stop statements respectively:





	start expr;


	stop expr;





where expr is a non-negative constant expression specifying the starting or stopping time in seconds. The following example





	start  20+30;


	stop  120;





set the starting time to t=50 and the stopping time to t=120, so the movie starts with the time variable “t” being 50 and stops at “t” being 120. The movie lasts 120-50=70 seconds.








3.13 COMMENTS





It is helpful to be able to put comments into a source program that can be read by humans but are invisible to the compiler. A comment in FMSL starts with two consecutive slashes // and may appear at any place in a script and will last to the end of the current physical line.

